5　極限，微分，積分，総和

関数の極限値は Limit を用いて計算する．式の微分には D を，積分には Integrate を用いる．ここで， D は derivative （微分）の頭文字で， Mathematica の組み込み関数名で略語が使用されている数少ない例の一つである．また，数列の総和は Sum を用いて求める．

■ 極限

■ 基本的な計算極限

Limit[expr, x->a] はxをaに近付けたときのexprの極限値を返す． $\lim_{x\to 1} \frac{x^2-1}{x-1}$ の値を求めてみる．

```
In[1]:= Limit[(x^2 - 1) / (x - 1), x -> 1]

Out[1]= 2
```

これを Cancel を用いて分数式を約分することにより確かめてみる．

```
In[2]:= Cancel[(x^2 - 1) / (x - 1)]

Out[2]= 1 + x
```

これに x=1 を代入すると 2 となる．

x→∞ における極限は x->Infinity を用いる．

```
In[3]:= Limit[(x + 2) / (3 x + 4), x -> Infinity]
```

Out[3]= $\frac{1}{3}$

■ 近づき方の指定

```
In[4]:= Plot[1 / x, {x, -1, 1}]
```




```
Out[4]= - Graphics -
```

$f(x) = \frac{1}{x}$のグラフは上図のように x=0 において不連続になっているので， x→0 における極限値は，下方から近づく場合と上方から近づく場合とで異なる．下方からの場合は Direction→1，上方からの場合は Direction→-1というオプションを付けることにより，この近づき方を指定できる．

```
In[5]:= Limit[1 / x, x -> 0, Direction -> 1]

Out[5]= -∞

In[6]:= Limit[1 / x, x -> 0, Direction -> -1]

Out[6]= ∞
```

練習問題： $\lim_{x\to 2} \frac{x-2}{\sqrt{x}-\sqrt{2}}$ の値を求めよ．

練習問題： $\lim_{n\to\infty} (1+\frac{1}{n})^n$ の値を求めよ．

練習問題： 関数 $f(x)$ の微分の定義はである．この定義どおりに $f(x) = x^3$, $f(x) = \sin x$ の微分を計算してみよ．

練習問題： 関数 $f(x)$ の微分の定義は $\lim_{h\to 0} \frac{f(x+h)-f(x)}{h}$ である．この定義どおりに $f(x) = x^3, f(x) = \sin x$ の微分を計算してみよ．

■ 微分

■ 基本的な計算

D[expr, x] は，式exprをxについて微分（偏微分）した式を返す．

```
In[7]:= D[Sin[x], x]

Out[7]= Cos[x]

In[8]:= D[x^n, x]
```

Out[8]= $n\, x^{-1+n}$

未定義の関数を微分すると関数名に ' が付いた関数になる．

```
In[9]:= D[f[x], x]

Out[9]= f′[x]
```

f を具体的な関数たとえば三角関数 Sin に置き換えると，それを微分した関数になる．

```
In[10]:= % /. f → Sin

Out[10]= Cos[x]
```

積や合成関数などに対しても，公式どおりに微分した式を返す．

```
In[11]:= D[f[x] g[x], x]
```

Out[11]= $g[x]\,f'[x] + f[x]\,g'[x]$

```
In[12]:= D[f[g[x]], x]
```

Out[12]= $f'[g[x]]\,g'[x]$

■ 高次の微分

D[expr, {x, n}] は，exprをxについてn回微分したもの（ n 次の導関数）を返す.

```
In[13]:= D[x^m, {x, 3}]
```

Out[13]= $(-2+m)\,(-1+m)\,m\,x^{-3+m}$

```
In[14]:= D[Sin[x], {x, 2}]
```

Out[14]= -Sin[x]

```
In[15]:= D[Sin[x], {x, n}]
```

Out[15]= $\mathrm{Sin}^{(n)}[x]$

$\mathrm{Sin}^{(n)}$ は Sin の n 次の導関数を意味する.

fの２階の導関数は f'' であるが，３階以上は $f^{(n)}$ で表される.

```
In[16]:= D[f[x], {x, 2}]
```

Out[16]= $f''[x]$

```
In[17]:= D[f[x], {x, 3}]
```

Out[17]= $f^{(3)}[x]$

D[expr, x, y] は，exprをxで微分したものをさらにyで微分して得られる関数を返す.

```
In[18]:= D[x^2 y^3, x, y]
```

Out[18]= $6\,x\,y^2$

■ 微分変数以外はすべて定数

D[expr, x] は，微分する変数x以外のシンボルはすべて定数とみなして微分をする.

```
In[19]:= D[a x + b y, x]
```

Out[19]= a

もしも，y が x の関数であるとして $a\,x+b\,y$ を微分したいのであれば，y の代わりに y[x] とすべきである.

```
In[20]:= D[a x + b y[x], x]
```

Out[20]= $a + b\,y'[x]$

> 練習問題： 多項式や sin, cos, log などから成る通常の関数では， x について微分した後に y について微分したものと， y について微分した後に x について微分したものとは一致する．このことを，， $f(x,\ y) = \log(x^2 + \sin y)$ に対して確かめよ．

積分

不定積分

`Integrate[expr, x]` は，式exprのxに関する不定積分を返す．

```
In[21]:= Integrate[x^3, x]
```

Out[21]= $\frac{x^4}{4}$

未定義の関数を積分すると，ただ積分したという記号がついただけの式を返す．

```
In[22]:= Integrate[f[x], x]
```

Out[22]= $\int f[x]\, dx$

関数 `f(x)` の不定積分は，通常 `F(x)+C` と積分定数 `C` を用いて表されるが， `Integrate` が返すものは，積分定数を省略してある．したがって，ある関数を微分したものをIntegrateにかけても元の関数に戻るとは限らない．

```
In[23]:= D[1 + x + x^2, x]
```

Out[23]= $1 + 2\,x$

```
In[24]:= Integrate[%, x]
```

Out[24]= $x + x^2$

> 練習問題： $f(x) = \sin\left(\frac{\cos x^2}{x \log x}\right) \cos x$ を微分した後，それを積分せよ．

定積分

`Integrate[expr, {x, a, b}]` は定積分を返す．

```
In[25]:= Integrate[f[x], {x, a, b}]
```

Out[25]= $\int_a^b f[x]\, dx$

```
In[26]:= Integrate[x^2, {x, 0, 1}]
```

Out[26]= $\frac{1}{3}$

積分区間の境界を無限大∞にとるには， `Infinity` を用いる．

```
In[27]:= Integrate[1 / x^2, {x, 1, Infinity}]
```

Out[27]= 1

```
In[28]:= Integrate[E^(-x^2), {x, -Infinity, Infinity}]
```

Out[28]= $\sqrt{\pi}$

■ 多重積分

複数の変数に対して積分することを多重積分という．

```
In[29]:= Integrate[f[x, y], x, y]
```

Out[29]= $\int\int f[x, y]\, dy\, dx$

```
In[30]:= Integrate[x y^2, x, y]
```

Out[30]= $\frac{x^2\, y^3}{6}$

■ 多重定積分

Integrate[expr, {x, xmin, xmax}, {y, ymin, ymax}] は xmin ≤ x ≤ xmax, ymin ≤ y ≤ ymax という領域上での定積分の値を返す．

```
In[31]:= Integrate[f[x, y], {x, xmin, xmax}, {y, ymin, ymax}]
```

General::spell1 : スペル間違いの可能性があります．新規シンボル"ymin"はすでにあるシンボル"xmin"に似ています． 詳細

General::spell1 : スペル間違いの可能性があります．新規シンボル"ymax"はすでにあるシンボル"xmax"に似ています． 詳細

Out[31]= $\int_{xmin}^{xmax}\int_{ymin}^{ymax} f[x, y]\, dy\, dx$

このメッセージは，ymin と xminが似ているがスペルミスではないのかという親切な警告ですので，無視してもかまいません．

$$\int_{xmin}^{xmax}\int_{ymin}^{ymax} f[x, y]\, dy\, dx$$

x と y の順序に注意．

```
In[32]:= Integrate[x y^2, {x, a, b}, {y, c, d}]
```

Out[32]= $\frac{1}{6}\,(a^2 - b^2)\,(c^3 - d^3)$

xmin,xmax,ymin,ymaxを定数とすれば積分領域は長方形となるが，そうでない場合でも積分できる．次の計算例は下図の領域すなわち $0<x<2\pi$, $\sin(x) < y < 2\sin(\frac{x}{2})$ においての積分である．



上の図は<<Graphics`FilledPlot`を実行してパッケージを読み込んだ後FilledPlot[{Sin[x],2Sin[x/2]},{x,0,2Pi},AxesLabel→{x,y}]とすれば得られる．

In[33]:= `Integrate[x y, {x, 0, 2 Pi}, {y, Sin[x], 2 Sin[x / 2]}]`

Out[33]= $\frac{3\pi^2}{2}$

練習問題： $f(x,\ y) = x\,y^2$ を次図のような領域上すなわち 0<x<1,0<y<1-x で積分せよ．



■ 数値積分

全ての関数の不定積分，定積分の値が求められるわけではない．原理的に不定積分が多項式や指数関数や三角関数などの既知の関数で表すことのできない場合もある．$\int \sin\cos x^2\,dx$ はそのような場合である．このような場合， Mathematica はそのままの式を返す．

In[34]:= `Integrate[Sin[Cos[x^2]], x]`

Out[34]= $\int \mathrm{Sin}[\mathrm{Cos}[x^2]]\,dx$

もちろん定積分も同様である．

In[35]:= `Integrate[Sin[Cos[x^2]], {x, 0, Pi}]`

Out[35]= $\int_0^\pi \mathrm{Sin}[\mathrm{Cos}[x^2]]\,dx$

しかしこのような場合でも，NIntegrate を用いて数値積分を行って定積分の近似値を求めることはできる．

In[36]:= `NIntegrate[Sin[Cos[x^2]], {x, 0, Pi}]`

Out[36]= 0.485846

練習問題： $\int_0^1 \log\sin(x^2)\,dx$ を数値積分せよ．

■ 総和

■ 基本的な計算

数列a[i]の総和は Sum[a[i], {i, imin, imax}] により求められる.

```
In[37]:= Sum[a[i], {i, 1, 5}]

Out[37]= a[1] + a[2] + a[3] + a[4] + a[5]
```

平方数の和 $1^2 + 2^2 + 3^2 + 4^2 + 5^2$ は次のようにして求められる.

```
In[38]:= Sum[i^2, {i, 1, 5}]

Out[38]= 55
```

n までの平方数の和 $1^2 + 2^2 + 3^2 + \cdots + n^2$ の一般式も求められる.

```
In[39]:= Sum[i^2, {i, 1, n}]
```

Out[39]= $\frac{1}{6}$ n (1 + n) (1 + 2 n)

■ ステップの指定

Sum[expr, {i, imin, imax, istep}] のようにステップを指定すると i を imin から imax まで istep ごとに動かして和をとる.

```
In[40]:= Sum[a[i], {i, 1, 10, 3}]

Out[40]= a[1] + a[4] + a[7] + a[10]
```

■ 多重和

Sum[expr, {i, imin, imax}, {j, jmin, jmax}] とすれば，複数個のパラメターによる和，多重和を求めることができる.

```
In[41]:= Sum[a[i, j], {i, 1, 2}, {j, 1, 3}]

Out[41]= a[1, 1] + a[1, 2] + a[1, 3] + a[2, 1] + a[2, 2] + a[2, 3]

In[42]:= Sum[i j^2, {i, 1, m}, {j, i, m}]
```

Out[42]= $\frac{1}{120}$ m (1 + m) (2 + m) (-1 + 9 m + 12 m^2)

練習問題： 1 から 2n+1 までの奇数の和を求めよ.

■ 演習問題

[5-1] $\lim_{x \to \frac{\pi}{6}} \frac{4 \sin^2 4 x + 3 \cos 6 x}{\sin 3 x - 2 \cos 2 x}$ を求めよ.

[5-2] $(f(x))^n$ を x について微分するとどのような関数になるか.

[5-3] y=sin x (0≤x≤π) のグラフを x 軸を中心に回転してできる回転体の体積を求めよ.

[5-4] $\sqrt{1 - x^2}$ を積分することにより，半円の面積を求めよ.

[5-8] $\sum_{k=1}^{n} (-1)^k \frac{k}{4 k^2 - 1}$ を求めよ.

■ この章で出てきた関数

`Limit[expr, x->a]`	極限値
`Limit[expr, x->a, Direction->±1]`	近づき方の指定
`D[expr, x]`	xで微分
`D[expr, {x, n}]`	n階の微分
`D[expr, x, y, ...]`	xで微分の後yで微分…
`Integrate[expr, x]`	xで不定積分
`Integrate[expr, {x, xmin, xmax}]`	xで定積分
`Integrate[expr, x, y, ...]`	多重積分
`Integrate[expr, {x, xmin, xmax}, {y, ymin, ymax}, ...]`	多重定積分
`NIntegrate[expr, {x, xmin, xmax}]`	数値積分
`Sum[expr, {i, imin, imax}]`	総和
`Sum[expr, {i, imin, imax, istep}]`	ステップを指定した総和
`Sum[expr, {i, imin, imax}, {j, jmin, jmax}, ...]`	多重総和